

第 14 课 精打细算——综合应用二

任务导航

生活中处处离不开计算，早晨算算还有多长时间上课，考试后算算自己能得多少分，买东西算算要花多少钱，开车旅行算算走哪条路近。

本课我们将为小学生制作一个口算练习小程序，如图 15.1 “百题赛”程序运行界面所示。




图 15.1 “百题赛”程序运行界面

Follow Me

制作“口算练习”程序。

任务分析：

- 运行程序后，可以通过单选按钮选择运算类型和计算范围。单击“进入考试”按钮后，弹出消息框，输入题目数量。单击“开始考试”按钮答题，对答题结果给与反馈。
- 本题可制作两个窗体，考试设置在第一个窗体中，答题在第二个窗体中。

准备：将“right.bmp”和“wrong.bmp”复制到“\第 3 单元\15 精打细算”中。

(1) 设计用户界面




图 15.2 程序设计界面

①设计窗体 1 (Form1) 界面

步骤 1：在工具箱中分别把两个框架 (Frame) 拖放到窗体上，调整到合适大小。

步骤 2：在两个框架中分别添加图像框控件数组如图 15.2 (a) 程序设计界面窗体 1 (Form1) 所示。

步骤 3：添加两个命令按钮 (CommandButton) 到窗体的合适位置。

②设计窗体 2 (Form2) 界面

步骤 1: 添加窗体 (Form2)。

步骤 2: 在窗体 2 上添加四个标签(Label), 一个文本框(TextBox), 三个命令按钮(CommandButton) 和一个图像框 (Image), 如图 15.2 (b) 程序设计界面窗体 2 所示。。

(2) 设置对象属性

根据表 15-1 设置对象属性。

15-1 (a) 窗体 1 对象属性设置表

| 对象 | 属性名 | 属性值 | 属性说明 |
|---------------------------------------|---------|---------|-------------------------------|
| Form1 (窗体) | Caption | 考试设置 | |
| Frame1 (框架 1) | Caption | 运算选择 | |
| Frame2 (框架 2) | Caption | 范围选择 | |
| OptionButton1 (单选按钮 1) 具有四个元素的控件数组 | 名称 | OptOper | |
| | Caption | | 分别为“+”、“-”、“*”、“/” |
| | Index | | 为别从 0 到 3 |
| OptionButton2 (单选按钮 2) 具有三个元素的控件数组 | 名称 | OptArea | |
| | Caption | | 分别为“10 以内”、“100 以内”、“1000 以内” |
| | Index | | 为别从 0 到 2 |
| Command1 (命令按钮 1) | 名称 | CmdEnd | |
| | Caption | 退出考试 | |
| Command2 (命令按钮 2) | 名称 | CmdTest | |
| | Caption | 进入考试 | |

15-1 (b) 窗体 2 对象属性设置表

| 对象 | 属性名 | 属性值 | 属性说明 |
|-------------------|-------------|----------------|------|
| Form2 (窗体) | Caption | 百题赛 | |
| Label1 (标签框 1) | 名称 | Lbl1Oper | |
| | Alignment | 1-Fixed Single | |
| | BackColor | &H00C0E0FF& | |
| | BorderStyle | 1-Fixed Single | |
| | Caption | | 空 |
| Label2 (标签框 2) | 名称 | LblQuestion | |
| | BackColor | &H80000013& | |
| | BorderStyle | 1-Fixed Single | |
| | Caption | | 空 |
| Label3 (标签框 3) | Caption | 题目 | |
| Label4 (标签框 4) | Caption | 输入答案 | |
| TextBox1 (文本框 1) | 名称 | | |
| | BackColor | &H00C0FFC0& | |
| | Text | | 空 |
| Command1 (命令按钮 1) | 名称 | CmdStart | |

| | | | |
|------------------|---------|---------|------------|
| | Caption | 开始考试 | |
| Command2(命令按钮 2) | 名称 | CmdSet | |
| | Caption | 重新开始 | |
| Command3(命令按钮 3) | 名称 | CmdExit | |
| | Caption | 考试结束 | |
| Command4(命令按钮 4) | 名称 | CmdNext | |
| | Caption | 下一道题 | |
| | Enable | False | |
| Image1(图像框 1) | 名称 | Image1 | |
| | Stretch | True | 图片随图像框大小变化 |

(3) 编写事件过程代码

①设计窗体 1 中的程序代码。窗体 1（考试设置）中设置的运算符和计算范围都要传送到 2（百题赛）中，所以要先在窗体 2 中定义需要的 Public 变量。

步骤 1：在窗体 2 中定义 Public 全局变量，用于存放运算符和计算范围。

```
Public oper As Integer      '存放所选运算符控件数组的序号
Public operation As String  '将所选运算符以字符串形式保存
Public area As Integer     '存放计算范围
```

智多星

变量的作用域

一个变量被定义后并不是在任何地方都能被引用的，每一个变量都有它的作用域。变量的有效范围就是变量的作用域。由于变量的有效范围不同，所以也就有了不同级别的变量之分。

(1) 局部变量

在一个过程内部（例如 Command1_Click）定义的变量称为局部变量。这个变量只能在定义它的过程内使用。

```
Private Sub Form1_Click()
    Dim Count As Integer
    .....
End Sub
```

(2) 窗体和模块级变量

如果一个窗体中的不同过程要使用同一个变量，这就需要在该窗体或模块内的过程外面定义一个变量，它在整个窗体或模块中有效，即其作用域为整个窗体或模块，本窗体或本模块内的所有过程都能访问它，这就是窗体或模块级的变量。

(3) 全局变量

全局变量可以被程序中任何一个模块和窗体访问。在窗体中不能定义全局变量，全局变量要在模块文件（.bas）中的声明部分用 Global 或 Public 关键字声明。

步骤 2：在窗体 1 的 Form_Load() 事件中，对两个控件数组进行初始化。如果用户使用默认选项，直接进入考试，则把默认运算符和计算范围赋值给窗体 2 相关变量。

```
Private Sub Form_Load()
    OptOper(0).Value = True    '默认选项为加法
    OptArea(1).Value = True    '默认计算范围是 100
```

```

Form2.operation = "+"      '将默认运算符赋值给窗体 2 中的变量
Form2.oper = 0            '将默认运算符的序号赋值给窗体 2 中的变量
Form2.area = 100         '将默认计算范围赋值给窗体 2 中的变量
End Sub

```

步骤 3: 将用户选择的运算符传递给窗体 2.

```

Private Sub OptOper_Click(Index As Integer)
    Form2.oper = Index
    Form2.operation = OptOper(Index).Caption
    '将用户所选运算符的 Caption 属性值传递给窗体 2
End Sub

```

步骤 4: 将用户选择的计算范围传递给窗体 2.

```

Private Sub OptArea_Click(Index As Integer)
    If Index = 0 Then Form2.area = 10
    If Index = 1 Then Form2.area = 100
    If Index = 2 Then Form2.area = 1000
End Sub

```

步骤 5: 运行、调试程序。

②设计窗体 2 中的程序代码

步骤 1: 定义程序中必要的变量 (窗体或模块级变量)。

```

Dim a(1 To 1000) As Integer, b(1 To 1000) As Integer, c(1 To 1000) As Integer
'定义三个数组, 分别存放算术题中的两个已知数和结果
Dim n As Integer      'n 存放题目数

```

步骤 2: 为“开始考试”按钮添加程序代码。

```

Private Sub CmdStart_Click()
    LblOper.Caption = operation      '将所选运算符显示出来
    n = InputBox("请输入考试题数", "考题设置", 2) '用户输入考题数量
    Randomize                        '初始化随机数生成器
    'oper 为 0 时做加法题, 为 1 时做减法题, 为 2 时做乘法题, 为 3 时做除法题
    Select Case oper
    Case 0
        For i = 1 To n
            a(i) = Fix(Rnd * area)      '随机生成一个加数
            b(i) = Fix(Rnd * (area - a(i))) '生成另一个加数
            c(i) = a(i) + b(i)          '答案
        Next
    Case 1
        For i = 1 To n
            a(i) = Fix(Rnd * area)      '随机生成一个被减数
            b(i) = Fix(Rnd * a(i))      '生成一个小于或等于被减数的随机数
            c(i) = a(i) - b(i)          '答案
        Next
    Case 2

```

```

For i = 1 To n
    a(i) = Fix(Rnd * area)           '乘法
    b(i) = Fix(Rnd * (area / a(i)))
    c(i) = a(i) * b(i)             '答案
Next
Case 3
For i = 1 To n
    a(i) = Fix(Rnd * area)           '除法
    b(i) = Fix(Rnd * (area / a(i))) 'b(i)存放除数
    c(i) = a(i) * b(i)
    x = a(i)
    a(i) = c(i)                     'a(i)存放被除数
    c(i) = x                         'c(i)存放商
Next
End Select
LblQuestion.Caption = Str(a(n)) + operation + Str(b(n))
'将第 n 题显示在相应标签框中, 出题顺序是从数组的第 n 项到第 1 项。
TxtResult.SetFocus                 '为输入答案的文本框设置焦点
CmdStart.Enabled = False           '“开始考试按钮”设置为不可用
CmdSet.Enabled = False             '“考试设置”为不可用
LblOper.Caption = "输入结果并以回车键结束"
End Sub

```

考试题的数目是由 InputBox() 来完成的, 如图 15.3 所示。




图 15.3 设置考试题数

步骤 3: 运行、调试程序。

DIY

试修改“百题赛”, 在窗体 1 中通过文本框设置题目的数量。

③设置“下一道题”按钮程序代码

```

Private Sub CmdNext_Click()
    TxtResult.Text = ""           '设置输入答案的文本框为空
    n = n - 1                     '设置下一道题的序号
    If n > 0 Then
        LblQuestion.Caption = Str(a(n)) + operation + Str(b(n)) '显示题目
        TxtResult.SetFocus
    Else
        LblOper.Caption = "考试结束!"
    End If
End Sub

```

```

        LblQuestion = ""
    End If
End Sub
运行、调试程序。
④判断答题正误
Private Sub TxtResult_KeyDown(KeyCode As Integer, Shift As Integer)
    If KeyCode = 13 Then ' 回车键
        If Val(TxtResult.Text) = c(n) Then
            LblOper.Caption = "答对了!"
            Image1.Picture = LoadPicture("right.bmp")
        Else
            LblOper.Caption = "答错了!"
            Image1.Picture = LoadPicture("wrong.bmp")
        End If
        CmdNext.Enabled = True
        CmdNext.SetFocus
    End If
End Sub

```

DIY

试完成“考试设置”和“考试结束”按钮的程序代码。

(4) 运行、调试并测试程序

运行、调试程序，测试程序出题、答题等功能。

(5) 保存程序

以文件名“百题赛”保存窗体文件和工程文件到“\第3单元\15百题赛”的文件夹中。

DIY

每当重新考试时，窗体2中的各个对象需要相应的初始化，如图15.4所示，请你来完成。




图 15.4 准备答题

知识着陆

1. 要根据实际作用域需求来定义变量。
 - (1) 在过程中只能定义局部变量。
 - (2) 在窗体或模块的声明部分可以定义窗体或模块级变量。
 - (3) 在模块的声明部分可以用 Public 定义全局变量。

视野拓展

变量

(1) 省略 As 子句的 Dim 语句将变量声明为变体类型

Dim 变量名

这种定义方式省略了 As 子句，VB 把这种格式声明的变量认为是 Variant 类型的数据。

例如：Dim a

(2) 用类型声明符声明变量类型

使用类型声明符（见表 1-1）隐含声明变量的数据类型。

为使程序具有比较好的可读性，建议在程序设计中采用第一种方式声明变

表 15-2 变量的类型声明符

| 类型 | 类型声明符 |
|-----------------|-------|
| Integer(整型) | % |
| Long(长整型) | & |
| Single (单精度实型) | ! |
| Double (双精度实型) | # |
| Currency (货币类型) | @ |
| String (字符串类型) | \$ |
| Variant (通用类型) | |

创作天地

1. 修改本课中的“百题赛”，使其具有记录并显示错题数目的功能。继续完善“百题赛”。




图 15.5 记录错题的数目

2. 试编程，制作一个简易计算器。

3. 试编程，将百题赛中所出题目都输出到窗体中。